


Lezioni di Ricerca Operativa 2

Dott. F. Carrabs


A.A. 2009/2010

Lezione in laboratorio:
- Esercizi

Lezione 3b:



Esempio: Miscela di Olii




Una fabbrica produce un tipo di cibo ottenuto dalla miscelazione di 5 differenti olii. Ci sono due olii di tipo vegetale (VEG1, VEG2) e tre olii di tipo non vegetale (OIL1, OIL2, OIL3). Ogni olio è inizialmente allo stato grezzo e deve essere raffinato. Ogni mese è possibile raffinare al massimo 200 tons di olii vegetali e 250 di olii non vegetali. Si suppone che non ci sia perdita di peso in questo processo di raffinamento e che il costo sia trascurabile. Ogni olio ha un costo ed una densità differente:

	VEG1	VEG2	OIL1	OIL2	OIL3
Costo (Euro/ton)	110	120	130	110	115
Densità (per ton)	8.8	6.1	2.0	4.2	5.0

Quando gli olii vengono mescolati la densità del prodotto finale cresce linearmente. La densità del prodotto finale deve essere compresa tra 3 e 6. Il prodotto finale viene venduto ad un prezzo di Euro 150 per ogni tonnellata.

COME PRODURRE IL PRODOTTO FINALE PER MASSIMIZZARE I PROFITTI NETTI?

Esempio : Miscela di Olii



var veg1 \geq 0;

var veg2 \geq 0;

var oil1 \geq 0;

var oil2 \geq 0;

var oil3 \geq 0;

var y \geq 0;

z*=17592.59259

maximize obj: $150*y - 110*veg1 - 120*veg2 - 130*oil1 - 110*oil2 - 115*oil3$;

subject to c1: $veg1 + veg2 + oil1 + oil2 + oil3 = y$;


s.t. c2: $veg1 + veg2 \leq 200$;

s.t. c3: $oil1 + oil2 + oil3 \leq 250$;

s.t. c4: $8.8*veg1 + 6.1*veg2 + 2*oil1 + 4.2*oil2 + 5*oil3 \leq 6*y$;

s.t. c5: $8.8*veg1 + 6.1*veg2 + 2*oil1 + 4.2*oil2 + 5*oil3 \geq 3*y$;

Esempio : Miscela di Olii



```
set OLII;  
set RISORSE;  
set DOMANDA;
```

```
var x{OLII}>=0;  
var y>=0;
```


$z^*=17592.59259$

```
param profitto;  
param costo{OLII};  
param b{RISORSE};  
param b2{DOMANDA};
```

```
param a{RISORSE,OLII};  
param a2{DOMANDA,OLII};
```

```
maximize obj: profitto*y - sum{i in OLII} costo[i]*x[i];  
subject to    c1 {i in RISORSE}: sum{j in OLII} a[i,j]*x[j] <= b[i];  
subject to    c2 {i in DOMANDA}: sum{j in OLII} a2[i,j]*x[j] <= y*b2[i];  
subject to    c3 : sum{i in OLII} x[i] = y;
```

Esempio : Miscela di Olii



set OLII:= VEG1 VEG2 OIL1 OIL2 OIL3;

set RISORSE:= Raff1 Raff2;

set DOMANDA:= D1 D2;

param profitto:= 150;

param costo:=

 VEG1 110

 VEG2 120

 OIL1 130

 OIL2 110

 OIL3 115;

param b:=

 Raff1 200

 Raff2 250;

param b2:=

 D1 6

 D2 -3;

param a: VEG1 VEG2 OIL1 OIL2 OIL3:=

 Raff1 1 1 0 0 0

 Raff2 0 0 1 1 1;

param a2: VEG1 VEG2 OIL1 OIL2 OIL3:=

 D1 8.8 6.1 2.0 4.2 5.0

 D2 -8.8 -6.1 -2.0 -4.2 -5.0;

```

set OLII;
set RISORSE;
set DOMANDA;


var x{OLII}>=0;
var y>=0;

param profitto;
param costo{OLII};
param b{RISORSE};
param b2{DOMANDA};

param a{RISORSE,OLII};
param a2{DOMANDA,OLII};

maximize obj: profitto*y - sum{i in OLII} costo[i]*x[i];
subject to    c1 {i in RISORSE}: sum{j in OLII} a[i,j]*x[j] <= b[i];
subject to    c2 {i in DOMANDA}: sum{j in OLII} a2[i,j]*x[j] <= y*b2[i];
subject to    c3 : sum{i in OLII} x[i] = y;
    
```

Esercizio: Curve Fitting



Una quantità y dipende da un'altra quantità x . Il seguente insieme di valori è stato osservato:


X	0.0	0.5	1.0	1.5	1.9	2.5	3.0	3.5	4.0
Y	1.0	0.9	0.7	1.5	2.0	2.4	3.2	2.0	2.7

X	4.5	5.0	5.5	6.0	6.6	7.0	7.6	8.5	9.0
Y	3.5	1.0	4.0	3.6	2.7	5.7	4.6	6.0	6.8

- 1) Determinare la retta $y = mx + q$ che minimizza la somma dei valori assoluti delle deviazioni dai dati osservati
- 2) Determinare la retta $y = mx + q$ che minimizza la massima deviazione dai dati osservati.

Risolvere il problema tramite l'ampl.

Esempio : Curve Fitting (minsum)



param begin;

param end > begin;

set Coordinate:= begin .. end;

param x{Coordinate};

param y{Coordinate};

var m;

var q;


var u{Coordinate}>=0;

var v{Coordinate}>=0;

minimize distance : sum{i in Coordinate} (u[i]+v[i]);

s.t. c1 {i in Coordinate}: $m \cdot x[i] + q + u[i] - v[i] = y[i]$;

Esempio : Curve Fitting (minsum)



```
param begin := 1;
param end   := 19;
```

```
param x:=
```

```
1 0.0
2 0.5
3 1.0
4 1.5
5 1.9
6 2.5
7 3.0
8 3.5
9 4.0
10 4.5
11 5.0
12 5.5
13 6.0
14 6.6
15 7.0
16 7.6
17 8.5
18 9.0
19 10.0;
```

```
param y:=
```

```
1 1.0
2 0.9
3 0.7
4 1.5
5 2.0
6 2.4
7 3.2
8 2.0
9 2.7
10 3.5
11 1.0
12 4.0
13 3.6
14 2.7
15 5.7
16 4.6
17 6.0
18 6.8
19 7.3;
```

```
param begin;
param end > begin;

set Coordinate:= begin .. end;

param x{Coordinate};
param y{Coordinate};


var m;
var q;
var u{Coordinate}>=0;
var v{Coordinate}>=0;

minimize distance : sum{i in Coordinate} (u[i]+v[i]);

s.t. c1 {i in Coordinate}: m*x[i]+q+u[i]-v[i]=y[i];
```

$z^* = 11.46625$

Esempio : Curve Fitting (minsum)



CPLEX 11.2.0: optimal solution; objective 11.46625


17 dual simplex iterations (0 in phase I)

$m = 0.6375$

$q = 0.58125$

	:	u	v :=
1		0,41875	0
2		0	0
3		0	0,51875
4		0	0,0375
5		0,2075	0
6		0,225	0
7		0,70625	0
8		0	0,8125
9		0	0,43125
10		0,05	0
11		0	2,76875
12		0	0,0875
13		0	0,80625
14		0	2,08875
15		0,65625	0
16		0	0,82625
17		0	0
18		0,48125	0
19		0,34375	0

Esempio : Curve Fitting (min violation)



```
param begin;  
param end > begin;
```

```
set Coordinate:= begin .. end;
```

```
param x{Coordinate};  
param y{Coordinate};
```


```
var m;  
var q;  
var u{Coordinate}>=0;  
var v{Coordinate}>=0;  
var z>=0;
```

```
minimize violation : z;
```

```
s.t. c1 {i in Coordinate}: m*x[i]+q+u[i]-v[i]=y[i];  
s.t. c2 {i in Coordinate}: z>=u[i];  
s.t. c3 {i in Coordinate}: z>=v[i];
```

$z^*=1.725$

Esempio : Curve Fitting



CPLEX 11.2.0: optimal solution; objective 1.725

11 dual simplex iterations (0 in phase I)

$m = 0.625$

$q = -0.4$

	:	u	v :=
1		1,4	0
2		0,9875	0
3		0,475	0
4		0,9625	0
5		1,2125	0
6		1,2375	0
7		1,725	0
8		0,2125	0
9		0,6	0
10		1,0875	0
11		0	1,725
12		0,9625	0
13		0,25	0
14		0	1,025
15		1,725	0
16		0,25	0
17		1,0875	0
18		1,575	0
19		1,45	0